# $Id: PKGBUILD 181879 2013-04-05 08:54:22Z svenstaro $
# Maintainer: Andrea Scarpino <andrea@archlinux.org>

pkgname=kdegraphics-okular
pkgver=4.10.2
pkgrel=1
pkgdesc='Document Viewer'
arch=('i686' 'x86_64')
url="http://kde.org/applications/graphics/okular/"
license=('GPL' 'LGPL' 'FDL')
groups=('kde' 'kdegraphics')
depends=('kdebase-runtime' 'qimageblitz' 'poppler-qt' 'chmlib' 'djvulibre'
         'ebook-tools' 'libspectre')
makedepends=('cmake' 'automoc4')
optdepends=('kdegraphics-mobipocket: mobipocket support')
install=${pkgname}.install
source=("http://download.kde.org/stable/${pkgver}/src/okular-${pkgver}.tar.xz")
sha1sums=('fcd0c725300c50ea04ec8e05b7d7035bf26afc79')

build() {
  cd "${srcdir}"
  mkdir build
  cd build
  cmake ../okular-${pkgver} \
    -DCMAKE_BUILD_TYPE=Release \
    -DKDE4_BUILD_TESTS=OFF \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DWITH_ActiveApp=OFF
  make
}

package() {
  cd "${srcdir}"/build
  make DESTDIR="${pkgdir}" install
}
